# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Thomas S Hatch <thatch45@gmail.com>
# Contributor: Jelle van der Waa <jelle@vdwaa.nl>

pkgname=python-pymongo
pkgver=4.3.3
pkgrel=2
pkgdesc='Python module for using MongoDB'
arch=('loong64' 'x86_64')
license=('Apache')
url='https://pypi.python.org/pypi/pymongo/'
depends=('python')
makedepends=('python-setuptools')
optdepends=('python-pyopenssl: for [ocsp]'
            'python-requests: for [ocsp]'
            'python-service-identity: for [ocsp]'
            'python-snappy: for [snappy]'
            'python-zstandard: for [zstd]'
            'python-dnspython: for [srv]')
source=("https://pypi.io/packages/source/p/pymongo/pymongo-$pkgver.tar.gz")
sha512sums=('f39b973b89495ca3bb9c0041ab750a6d31ac0d013e9a3eda63c32a6db17f8d46a24fd97456840274af1720d3ac2e05e137530d96b1f9a02feb89659cda821d61')

build() {
  cd pymongo-$pkgver
  python setup.py build
}

package() {
  cd pymongo-$pkgver
  python setup.py install --root="$pkgdir" --skip-build --optimize=1
}

# vim:set ts=2 sw=2 et:
